J. Fac. Appl. Biol. Sci.,
Hiroshima Univ. (1979), 18 : 11 − 30

Determination of the Empirical Rate Equation for the Chemical and Physical Transformations of Foods

Kiyoshi KUBOTA

Faculty of Applied Biological Science, Hiroshima University, Fukuyama

Received March 6, 1979
(Figs. 1–8, Tables 1–4, Appendix)

INTRODUCTION

In order to design various chemical and physical transforming apparatuses such as for the chemical reactions, drying, soaking, cooking, extracting, cooling, heating and so on of food materials, it is necessary to determine first the chemical and physical transforming rate equations from the experimental data of chemical reaction, drying, cooking and so on of foods.

In previous papers, we have studied the drying rate equations of agar gel and carrot [1)], the soaking and cooking rate equations of rice [2–6)], *udon* and *kishimen*[2)], and potato and sweet potato [7)], and so on[8,9)].

As the types of the chemical and physical transforming rate equations, we can consider two types. One type is a theoretical or semi-theoretical rate equation based on various models such as the drying-shell model[1)] and the soaking-shell model[2)]. This type postulates the conclusions of the previous papers. The other type is an empirical formula[5–8)]. This one is useful, unless the chemical and physical transforming mechanism can be analyzed and their transforming model can be postulated approximately. In this paper, we studied the derivation method of the best empirical rate equations for the various chemical and physical transformations of food materials.

EMPIRICAL RATE EQUATIONS

The degree or extents y (example of unit: g) of chemical reaction, drying, soaking, cooking, extracting, cooling, heating and so on of food materials can be expressed as the changing values of a property such as weight, reological property, temperature and so on of foods. In general, the experimental data of the chemical and physical trnasformations are obtained as the relationship of y vs. θ, where θ (example of unit: min) is the chemical and physical transforming time of foods.

Most of the chemical and physical transformations of foods have two boundary values y_0 and y_e correlated to the initial and equilibrium states, respectively. The values of y differ greatly by used units, the species of observed property and the limited values of the initial and equilibrium states which are not values kept constant throughout the various experimental conditions. The dimensionless expression $x(-)$ as in the following equation is convenient for the understanding of the chemical and transforming ratio.

$$x = |(y - y_0)/(y_e - y_0)| \tag{1}$$

The values of x vary from 0 to 1 with the changes of y from y_e to y_e. When the measures of the reactant concentration on the chemical reaction, the sample weight on the drying, or the temperature on the cooling of foods are taken as the degree of chemical and physical transformations, the values of y_0 are larger than y_e. On the other hand, when the product concentration on the reaction, the sample weight on the soaking, or the temperature on the heating of foods are measured, the values of y_e are larger than y_0.

Most of the relationships of x vs. θ show a monotonous smooth or a S-shape curve. The empirical formula of the rate equation may be postulated as follows:

$$dx/d\theta = k_{n,\alpha}(1-x)^n(x+\alpha) \tag{2}$$

where, k_n, α (example of unit: min^{-1}), $n(-)$ and α $(-)$ are the rate parameters which can be obtained from the experimental data of x vs. θ. When the relationships of x vs. θ do not show a S-shape curve, the rate equation has to be postulated simply as follows:

$$dx/d\theta = k_n(1-x)^n \tag{3}$$

where, k_n (example of unit: min^{-1}) and $n(-)$ are the rate parameters which can be obtained.

Most of the data have considerable scatters, then the rate parameters in the rate equations show some interactions. When the number of the rate parameters increases, the interaction too appreciably increases. Therefore, the two rate parameter's equation such as Eq.(3) is more useful than the three rate parameter's equation such as Eq.(2). If the values of α in Eq.(2) is large, Eq.(3) excepted α is useful. When the values of α is small, the following two rate parameter's equation assumed n=1 is useful.

$$dx/d\theta = k_\alpha(1-x)(x+\alpha) \tag{4}$$

As a second step, it is necessary to replace the rate equation of Eq.(2) by that of Eq.(3) or (4). The first step which is necessary as an earlier step is the calculation of the rate parameters $k_{n,\alpha}$, n and α in Eq.(2). The values of k_n and k_α in Eqs.(3) and (4) change appreciably with a slight variation of the values of n and α, respectively. As the third step, it is necessary to determine the average values of n and α and to fixing by those approximate values. After these two steps have been attained, we can consider the Arrhenius and so on relationships on the values of k_n and k_α.

CALCULATION OF INITIAL RATE PARAMETERS

The experimental data are generally obtained as the integral data of x vs. θ. The rate parameters k_n and n in Eq.(3) may be obtained by a linear least square method, from the data x vs. θ, using the following equations deformed Eq.(3), but the rate parameters $k_{n,\alpha}$, n and α in Eq.(2) can not be obtained by the same method.

$$\ln(dx/d\theta) = \ln(k_n) + n\ \ln(1-x) \tag{5}$$

Most of the data show considerable scatter, cosequently the derivative values of $dx/d\theta$ in Eq.(5) cannot be obtained reliably find from the data x vs. θ. This differential analysis is in general not better than the following integral analysis.

For an analytical integral analysis, Eqs.(3) and (4) become as follow:

From Eq.(3) :

$$k_n = ((1/(1-x))^{n-1} - 1)/((n-1)\theta) \quad \text{for } n \neq 1.0 \tag{6}$$

$$k_{n=1} = -\ln(1-x)/\theta \quad \text{for } n = 1.0 \tag{7}$$

From Eq.(4) :

$$k_\alpha = \frac{1}{(1+\alpha)\theta} \ln \frac{x+\alpha}{(1-x)\alpha} \tag{8}$$

However, Eq.(2) cannot be integrated analytically, therefore we must use a numerical integral analysis. For carrying out the calculations of Eqs.(6)~(8), the assumptions of the values of n and α are required. The analytical integral analysis is really complicated. The numerical integral analysis used a digital electric computer which was most successful in various analysis.

Thus, for the preparation of the program on the determination of the empirical rate equations, Eqs.(2), (3) and (4) were integrated numerically using the Runge-Kutta-Gill method. The rate parameters $k_{n,\alpha}$, k_n, k_α, n and α were calculated by a non-linear least square method[10−12](we used the digital electric computer "HITAC 8700−OS7" in the Computation Center of Hiroshima Univ.).

The values of the following standard deviation $\sigma(-)$ for the valiable x were minimized.

$$\sigma = \left(\sum_{i=1}^{N} (x_{obs} - x_{cal})_i^2 / N\right)^{1/2} \tag{9}$$

where, x_{obs} and x_{cal} are the observed and calculated values of x, and N is the total number of the experimental points.

In the numerical integral analysis which used the non-linear least square method, the initial values of the rate parameter are required. The initial values of the rate parameter k_n in Eq.(3) can be calculated by Eqs.(6) and (7). For the cases of unknown n (second step), the initial values of k_n were calculated by Eq.(7) assuming n=1.0, and for the fixed n=0.5, 1.0 and so on (third setp), the values of k_n were calculated by Eqs.(6) and (7).

The initial values of $k_{n,\alpha}$ which assuming n=1 in Eq.(2), and k_α and α in Eq. (4) can be calculated by using the following obtained equations. In Eq.(4) fixing n=1 in Eq.(2), the deformation point of the S-shape curvatures may be given by (x_d, θ_d). Then, the values of x_d and θ_d can be obtained fron the following treatments:

$$(d^2x/d\theta^2)_{x=x_d} = k_\alpha(1-\alpha-2x_d) = 0$$

$$\therefore \; x_d = (1/2)(1-\alpha) \tag{10}$$

Combining Eqs.(8) and (10), we obtain :

$$\theta_d=(1/((1+\alpha)k_\alpha))\ln(1/\alpha) \tag{11}$$

The tangent line which touches the S-shape curve at point (x_d, θ_d) may be given by

$$x=a\theta+b \tag{12}$$

The values of a and b can be obtained as following :

$$a=(dx/d\theta)_{x=x_d}=k_\alpha(1-x_d)(x_d+\alpha)=(1/4)k_\alpha(1+\alpha)^2 \tag{13}$$

$$b=x_d-a\theta_d=(1/2)(1-\alpha)+(1/4)(1+\alpha)\ln\alpha \tag{14}$$

For $\alpha \ll 1.0$: $a=(1/4)k\alpha$, $b=(1/2)+(1/4)\ln\alpha$ (15)

For $\alpha = 1.0$: $a=k_\alpha$, $b=0$ (16)

The initial values of the rate parameters $k_{n,\alpha}$, k_α and α in Eqs.(2) and (4) were calculated from the calculated values of a and b in Eq.(12) using the linear least square method[11-13]. For the calculations of a and b in Eq.(12), the data in the limitted range of $0.2 \leqq x \leqq 0.8$ are used. For the calculated results of $b<0$ and $b \geqq 0$ (first and second steps), the initial values of k_α and α were calculated from a and b using Eqs.(15) and (16), respectively. For the fixed α=0.5, 0.1 and so on (third step), the values of k_α were calculated by Eqs. (13) and (14).

DETERMINATION OF EMPIRICAL RATE EQUATIONS



Fig. 1. Flow chart of main program for the determination of empirical rate equation (Cont. to Fig. 3).

From the experimental data of x vs. θ, the rate parameters $k_{n,\alpha}$, n and α in Eq. (2) must be calculated as the first step calculation by a non-linear least square method using a digital electric computer. The flow chart for the first step calculation is shown in Fig. 1.

For the reason descrived above, Eq. (2) must be simplified to Eq.(3) of (4) which have a lesser number of rate parameters for the second step calculation. For the higher average values of $\alpha(\alpha_{av} \geqq 0.75)$, Eq.(3) can be used as the next step rate equation, and for the lower values $(\alpha_{av} < 0.75)$, Eq.(4) can be used. The reason why the value of 0.75 was used is shown in Fig. 2. The curves in Fig. 2 are obtained from the following equations alternating Eqs.(6)~(8).



Fig. 2. Relations between the dimensionless extent and the transforming time related rate parameters for various values of n or α.
Calculated results of solid lines (x vs. $k_n\theta$) : for Eq. (3)
From upper line : n=0.5, 1.0, 1.5 and 2.0
where, a=1.17, 1.39, 1.66 and 2.00
Calculated results of broken lines (x vs. $k_\alpha\theta$) : for Eq.(4)
From upper line: α=0.001, 0.01, 0.1 and 0.5
where, a=13.81, 9.25, 4.25 and 1.85

From Eq.(3):

$$x=1-(1+k_n\theta(n-1))^{1/(1-n)} \quad \text{for} \quad n\neq 1.0 \tag{17}$$

$$x=1-\exp(-k_{n=1}\theta) \quad \text{for} \quad n=1.0 \tag{18}$$

From Eq.(4) :

$$x=\frac{\alpha(\exp((1+\alpha)k_\alpha\theta)-1)}{\alpha\exp((1+\alpha)k_\alpha\theta)+1} \tag{19}$$

From the obtained curves in Fig. 2, it was found that the results of Eq.(3) used lower values of n approach where the results of Eq.(4) used higher values of α. The curves of n=0.5 and α=0.5 are almost similar.

For the reason descrived above, Eqs.(3) and (4) must be simplified to the rate equations fixing n and α as the third step calculation, respectively. From the calculation of the average values of n and α in a series of data, the most approximate fixing values of n and α can be obtained. For the third step calculation of Eq.(3), we used n=0.5, 1.0, 1.5 and 2.0 as the fixing values in the cases of $n_{av} < 0.75$, $0.95 \leqq n_{av} < 1.25$, $1.25 \leqq n_{av} < 1.75$ and $1.75 \leqq n_{av}$, respectively. For the calculation of Eq.(4), α=0.5, 0.1, 0.01 and 0.001 were used in the cases of $\alpha_{av} \geqq 0.25$, $0.25 > \alpha_{av} \geqq 0.05$, $0.05 > \alpha_{av} \geqq 0.005$ and $0.005 > \alpha_{av}$, respectively. The flow chart for these calculations is shown in Fig. 3. The flow chart of relationship between main program and subroutines is shown in Fig. 4, and the practical program is shown in Appendix. The calculated results in Appendix are as we shall show later.



Fig. 3. Flow chart of main program for the determination of empirical rate equation (Cont. from Fig. 2).



Fig. 4. Flow chart showed the relationship between main program and subroutines.

EXAMPLES OF CALCULATION

As data of the examples of calculations, the relationships of y (example of unit: g) vs. θ (example of unit: min) were estimated by the following equations alternating Eqs.(1)

Table 1. Rate parameters estimated dada.

k_n (min^{-1}) and $n(-)$ for Eq.(3)						k_α(min^{-1}) and $\alpha(-)$ for Eq.(4)					
Run	k_n	n	Run	k_n	n	Run	k_α	α	Run	k_α	α
1−1	0.060	0.5	3−1	0.30	1.5	5−1	0.12	0.5	7−1	0.30	0.01
2	0.050	0.5	2	0.25	1.5	2	0.10	0.5	2	0.25	0.01
3	0.040	0.5	3	0.20	1.5	3	0.08	0.5	3	0.20	0.01
4	0.030	0.5	4	0.15	1.5	4	0.06	0.5	4	0.15	0.01
2−1	0.12	1.0	4−1	0.80	2.0	6−1	0.20	0.1	8−1	0.40	0.001
2	0.10	1.0	2	0.60	2.0	2	0.15	0.1	2	0.30	0.001
3	0.08	1.0	3	0.50	2.0	3	0.12	0.1	3	0.25	0.001
4	0.06	1.0	4	0.40	2.0	4	0.10	0.1	4	0.20	0.001

and (6)~(8) and using the rate parameters from Table 1. The number of runs in one series of data with the same value of n or α was estimated to be four in each case.

From Eq.(1):

$$y = y_0 + (y_e - y_0)\,x \tag{20}$$

From Eq.(3) :

$$\theta = ((1/(1-x))^{n-1} - 1)/((n-1)\,k_n) \qquad \text{for} \quad n \neq 1.0 \tag{21}$$

$$\theta = -\ln(1-x)/k_{n=1} \qquad \text{for} \quad n = 1.0 \tag{22}$$

From Eq. (4) :

$$\theta = \frac{1}{(1+\alpha)\,k_\alpha} \ln \frac{x+\alpha}{(1-x)\,\alpha} \tag{23}$$

The data were given at the experimental points of x=0.05, 0.1, 0.2, . . . 0.8, 0.9, 0.95 (–) as shown later in Table 3. The number of data points was eleven. We used the values of θ calculated with the significant figure of three from the values of x. The values of y which were used as the original data were calculated with the significant figure of four from the values of x, using y_0=1.500 and y_e=1.000. The estimated data are shown by using sphericals sysmbol in Figs. 5~8.



Fig. 5. Relations between the dimensionless extent and the transforming time.
From upper solid line : for Run 1–1, 1–2, 1–3 and 1–4
○: estimated data, – : calculated results
From upper broken line : for Run 3–1, 3–2, 3–3 and 3–4
△: estimated data, – : calculated results



Fig. 6. Relations between the dimensionless extent and the transforming time.
From upper solid line : for Run 2–1, 2–2, 2–3 and 2–4
○: estimated data, – : calculated results
From upper broken line : for Run 4–1, 4–2, 4–3 and 4–4
△: estimated data, – : calculated results



Fig. 7. Relations between the dimensionless extent and the transforming time.
From upper solid line : for Run 8–1, 8–2, 8–3 and 8–4
○: estimated data, – : calculated results
From upper broken line : for Run 6–1, 6–2, 6–3 and 6–4
△: estimated data, – : calculated results



Fig. 8. Relations between the dimensionless extent and the transforming time.
From upper solid line : for Run 7–1, 7–2, 7–3 and 7–4
○: estimated data, – : calculated results
From upper broken line : for Run 5–1, 5–2, 5–3 and 5–4
△: estimated data, – : calculated results

The rate parameters in Eqs.(2)~(4) were calculated from the above estimated data using the program in Appendix. A part of the calculated results in shown in Tables 2~4. As for the calculated results of the rate parameters shown in Table 2, the best empirical rate equation may be determined as the rate equation for the third step. The values of the rate parameters k_n or k_α of the third step fixing both n and α in a series of data may be correlated to the experimental conditions such as the chemical and physical transforming temperature and so on. The calculated results compared to the estimated data are illustrated by solid lines in Figs. 5~8. The example of the calculated result

Table 2. Initial and calculated values of rate parameters.

For Run 2–1

Step	Initial values				Calculated values			
	k(min^{-1})	n(–)	α(–)	δ(–)	k(min^{-1})	n(–)	α(–)	δ(–)
1	0.5*	1.0	1.0	0.386	0.0876	1.28	1.21	0.00984
2	0.120**	1.0	–	0.000392	0.120	1.00	–	0.000383
3	0.120**	(1.0)	–	0.000392	0.120	(1.0)	–	0.000383

For Run 7–1

Step	Initial values				Calculated values			
	k(min^{-1})	n(–)	α(–)	δ(–)	k(min^{-1})	n(–)	α(–)	δ(–)
1	0.267*	1.0	0.0150	0.0255	0.298	0.993	0.0102	0.00330
2	0.267***	(1.0)	0.0150	0.0255	0.300	(1.0)	0.00994	0.00333
3	0.315***	(1.0)	(0.01)	0.0403	0.300	(1.0)	(0.01)	0.00333

( ): fixed values, *, ** and ***: k is $k_{n,\alpha}$ in Eq. (2), k_n in Eq.(3) and k_α in Eq.(4), respectively.

Table 3. Comparisons of estimated data and calculated values of dimensionless extent $x(-)$.

x_{obs} (−)	y_{obs} (g)	For Run 2−1 θ (min)	$x_{cal}(-)$ Step 1	Step 2	Step 3	For Run 7−1 θ (min)	$x_{cal}(-)$ Step 1	Step 2	Step 3
0	1.500	0	0	0	0	0	0	0	0
0.05	1.475	0.427	0.045	0.050	0.050	6.08	0.050	0.050	0.050
0.10	1.450	0.878	0.091	0.100	0.100	6.26	0.100	0.100	0.100
0.20	1.400	1.86	0.186	0.200	0.200	10.8	0.200	0.200	0.200
0.30	1.350	2.97	0.285	0.300	0.300	12.5	0.298	0.298	0.299
0.40	1.300	4.26	0.388	0.400	0.400	13.9	0.396	0.396	0.396
0.50	1.250	5.78	0.492	0.500	0.500	15.4	0.509	0.509	0.509
0.60	1.200	7.64	0.596	0.600	0.600	16.6	0.599	0.600	0.600
0.70	1.150	10.0	0.697	0.699	0.699	18.6	0.696	0.697	0.606
0.80	1.100	13.4	0.796	0.800	0.800	19.8	0.799	0.799	0.798
0.90	1.050	19.2	0.890	0.900	0.900	22.5	0.901	0.900	0.900
0.95	1.025	25.0	0.936	0.950	0.950	25.0	0.951	0.951	0.950
1.00	1.000	50.0	1.00	1.00	50.0	50.0	1.00	1.00	1.00

Table 4. Calculated values of rate parameters.

Run	Step	$k(\text{min}^{-1})$	$n(-)$	$\alpha(-)$	$\delta(-)$	Run	Step	$k(\text{min}^{-1})$	$n(-)$	$\alpha(-)$	$\delta(-)$
1−1	1	0.0569	0.842	0.888	0.0157	5−1	1	0.119	0.998	0.503	0.000784
	2	0.0600	0.500	−	0.000610		2	0.120	(1.0)	0.500	0.000788
	3	0.0600	(0.5)	−	0.000610		3	0.120	(1.0)	(0.5)	0.000788
1−4	1	0.0292	0.852	0.849	0.0200	5−4	1	0.0606	1.00	0.495	0.000270
	2	0.0300	0.502	−	0.000355		2	0.0601	(1.0)	0.499	0.000293
	3	0.0300	(0.5)	−	0.000400		3	0.0600	(1.0)	(0.5)	0.000298
2−1		shown in Table 3				6−1	1	0.201	1.00	0.0993	0.000330
2−4	1	0.0474	1.30	1.08	0.0135		2	0.201	(1.0)	0.0994	0.000331
	2	0.0599	0.999	−	0.000435		3	0.200	(1.0)	(0.1)	0.000396
	3	0.0599	(1.0)	−	0.000442	6−4	1	0.100	1.00	0.0999	0.000575
3−1	1	1.51	1.71	1.82	0.00662		2	0.100	(1.0)	0.100	0.000576
	2	0.301	1.50	−	0.000307		3	0.100	(1.0)	(0.1)	0.000578
	3	0.300	(1.5)	−	0.000343	7−1		shown in Table 3			
3−4	1	0.0954	1.76	1.41	0.00794	7−4	1	0.148	0.986	0.0103	0.00268
	2	0.150	1.50	−	0.000352		2	0.150	(1.0)	0.00991	0.00281
	3	0.150	(1.5)	−	0.000416		3	0.150	(1.0)	(0.01)	0.00282
4−1	1	0.298	2.18	2.57	0.00507	8−1	1	0.397	0.989	0.00104	0.00238
	2	0.814	2.02	−	0.00247		2	0.400	(1.0)	0.000986	0.00247
	3	0.806	(2.0)	−	0.00276		3	0.400	(1.0)	(0.001)	0.00249
4−4	1	0.185	2.21	2.03	0.00625	8−4	1	0.201	1.00	0.000962	0.000538
	2	0.407	2.02	−	0.00244		2	0.201	(1.0)	0.000980	0.000591
	3	0.403	(2.0)	−	0.00269		3	0.200	(1.0)	(0.001)	0.000755

(   ) etc. : same to Table 3.

shown in Appendix is the result of Run 7−1 in which the number of runs in a series of data had decreased simply into one.

If the values of n and α cannot be obtained as constant values in one series of data, the parameters must be correlated to the experimental conditions such as temperature and so on, too. When the range of conditions is wide, the values of k_n or k_α cannot be theoretically correlated well to all the conditions. If the values of n and α cannot be obtained as constant values in one series of data, it is preferable to reduce the range of conditions in one series of data.

The method and the computer program for the determination of the empirical rate equations shown in this paper are available for a series of data giving a monotonous smooth curve or a S-shape surve from which the values of n and α can be obtained as constant values.

This computer program can be used not only for the rate equations of the various transformations of foods, but also for the over-all rate equations of the thermal and catalytic cracking of hydrocarbons[16-18], and so on. For the heat transfer phenomena, the values of y can be obtained as the values of the temperature of the sample, and the Fourie's equation may be taken from Eq.(4) fixing n=1. One the other hand, for the mass transfer phenomena, the value of y can be obtained as the concentration, and the Fick's equation may be taken, too.

RESULTS

The method and the computer program for determining the empirical rate equations of the chemical and phsical transformations of foods were investigated, and then the examples of the calculations using estimated data were demonstrated.

Since the mechanism of the transforming of foods is really complicated, the program for the determination of the empirical rate equations shown in this paper may be very useful in these cases.

SUMMARY

In order to design and automatically control various chemical and physical transforming apparatuses such as chemical reaction, drying, cooking, extracting apparatuses and so on of food materials, it is required to formulate first the best transforming rate equations. Most of the transforming mechanisms of foods can be analyzed, but a theoretical rate equation can not be obtained. In these cases, the empirical formula of rate equation must be used. We investigated the method and the computer programs in order to determine the empirical rate equations. The method and the computer program of the determination of the empirical rate equations shown in this paper are available for the data giving a monotonous smooth curve or a S-shape curve. The program shown in this paper can be used not only for the determination of the empirical rate equation of the various foods, but also for the empirical rate equations of other materials.

NOTATIONS

a and b : constants in Eq.(12), (example: min^{-1}) and (–)

$k_{n,\alpha}$, k_n and k_α : rate parameters in Eqs.(2) and (4), respectively, (example: min^{-1})

N	:	total number of experimental points, (−)
n	:	rate parameter in Eqs.(2) and (3), (−)
x	:	dimensionless expression of y expressed by Eq.(1), (−)
y	:	degree or extent of chemical and physical transformations, (example: g)
α	:	rate parameter in Eqs.(2) and (4), (−)
θ	:	chemical and physical transforming time, (example: min)
σ	:	standard deviation by Eq. (9), (−)

Subscripts ;

av	:	average value in a series of data
d	:	deformation point of S-shape curve
0 and e	:	initial and equilibrium states
obs and cal	:	observed and calculated values

REFERENCES

1) KUBOTA, K., KOBATAKE, H., SUZUKI, K. and HOSAKA, H. : *J. Fac. Fish. Anim. Husb., Hiroshima Univ.*, **16**, 123–130 (1977).
2) KUBOTA, K., SUZUKI, K., HOSAKA, H., HIRONAKA, K. and AKI, M. : *ibid*, **15**, 135–149 (1976).
3) SUSUKI, K., KUBOTA, K., OMICHI, M. and HOSAKA, H. : *J. Food Sci.*, **41**, 1180–1183 (1976).
4) SUZUKI, K., AKI, M., KUBOTA, K. and HOSAKA, H. : *ibid*, **42**, 1545–1548 (1977).
5) KUBOTA, K., HIRONAKA, K., SUZUKI, K. and HOSAKA, H. : *Nippon Shokuhin Kogyo Gakkaishi*, **25**, 251–256 (1978).
6) KUBOTA, K., HIRONAKA K., HOSOKAWA, Y., SUZUKI, K. and HOSAKA, H. : *ibid*, **25**, 641–644 (1978).
7) KUBOTA, K., OSHITA, K., HOSOKAWA, Y., SUZUKI, K. and HOSAKA, H. : *J. Fac. Fish, Anim, Husb., Hiroshima Univ.*, **17**, 97–106(1978).
8) KUBOTA, K., HOSOKAWA, Y., SUZUKI, K. and HOSAKA, H. : *J. Food Sci.*, preparing.
9) KUBOTA, K., HOSOKAWA, Y., SUZUKI, K. and HOSAKA, H. : *New Food Industry*, **21** (3), 33–48, **21**(4), 36–47 (1979).
10) KUBOTA, K. and MORITA, N. : *Computation Center News of Nagoya Univ.*, **4**(4), 318–326 (1973).
11) HOSAKA, H., KUBOTA, K. and SUZUKI, K. : Shokuhin Kogaku, p.170–179, Kyoritsu Shuppan Co., Ltd., Tokyo (1975).
12) KUBOTA, K., SUZUKI, K. and HOSAKA, H. : *Shokuhin Kogyo*, **20** (16), 60–73 (1977).
13) KUBOTA, K. and MORITA, N. : *Computation Center News of Nagoya Univ.*, **4**(3), 234–237 (1973).
14) KUBOTA, K. SUZUKI, K. and HIRONAKA, K.: *Shokuhin Kogyo*, **20**(12), 66–72 (1977).
15) KUBOTA, K. HOSAKA, H. and AKI, M. : *ibid*, **20**(14), 49–55 (1977).
16) KUBOTA, K. : *Nippon Kagaku Kaishi*, No.7, 1276–1282 (1974).
17) KUBOTA, K. : *ibid*, No.8, 1393–1398 (1974).
18) KUBOTA, K. and TESHIMA, H. : *ibid*, No.12, 2391–2395 (1974).

APPENDIX

```
C      MAIN PROGRAM,  BY KIYOSHI KUBOTA.
C      ESTIMATION OF RATE EQUATIONS,
C      BY NONLINEAR LEAST SQUARE METHOD.
C      YB(MY,N,M)=,   DEPENDENT VARIABLES.
C      XB(MX,N,M)=,   INDEPENDENT VARIABLES.
C      MY,MX=,  NUMBER OF DEPENDENT AND INDEPENDENT VARIABLES.
C      N,M=,  NUMBER OF EXPERIMENTAL POINT AND RUN.
C      AB(L,K,M),SD(K)=,  PARAMETER AND STANDARD DEVIATION.
C      L,K=,  NUMBER OF PARAMETER AND ITERATION.
C      YC(MY,N,K)=,  CALCULATED VALUES OF DIPENDENT VARIABLES.
       COMMON AN,AM
       DIMENSION YBA(2,10),YB(2,25,10),YBE(2,10),XBA(1,10),XB(1,25,10)
      1,XBE(1,10),X(1,25),XA(1),XE(1),HX(1),Y(2,25),W(2,25),DYA(2,1)
      2,YA(2),YE(2),A(3,21),HM(21),HS(21),SD(21),YC(2,25,21),DYC(2,25,21)
      3,AB(3,10),YCS(2,25,4),DD(3,4),DYY(2),YY(2),AA(3),Q(2),DYS(2),YS(2)
      4,YYY(25),XX(2,25),WW(25),AAA(2),YYC(25),NM(10)
       DATA HAA/0.0005/,HM(1),HS(1)/2*1.0/,HMM,HSS/2*0.5/,HST/0.0001/
      1,EPS/1.0E-50/,MYT/2/,MXT/1/,NT/25/,MT/10/,LT/3/,L1/4/,KT/20/
      2,K1/21/,NNT/25/,LLT/2/
       EXTERNAL DIFEQA,DIFEQB,DIFEQC,DIFEQD,DIFEQE
   100 READ(5,10) M
       WRITE(6,50) M
       IF(M) 105,105,110
   110 IM=0
   115 IM=IM+1
       IF(M-IM) 120,125,125
   125 READ(5,10) MY,MX,N
       READ(5,15) (YBA(JY,IM),(YB(JY,I,IM),I=1,N),YBE(JY,IM),JY=1,MY)
       READ(5,15) (XBA(JX,IM),(XB(JX,I,IM),I=1,N),XBE(JX,IM),JX=1,MX)
       WRITE(6,51) MY,MX,N
       WRITE(6,52) (YBA(JY,IM),(YB(JY,I,IM),I=1,N),YBE(JY,IM),JY=1,MY)
       WRITE(6,53) (XBA(JX,IM),(XB(JX,I,IM),I=1,N),XBE(JX,IM),JX=1,MY)
       NM(IM)=N
C      CALCULATION OF Y(MY) CONVERTED RATIO.
       DO 130 JY=1,MY
       DO 135 I=1,N
   135 YB(JY,I,IM)=ABS((YBA(JY,IM)-YB(JY,I,IM))/(YBA(JY,IM)-YBE(JY,IM)))
       YBA(JY,IM)=0.0
   130 YBE(JY,IM)=1.0
       WRITE(6,52) (YBA(JY,IM),(YB(JY,I,IM),I=1,N),YBE(JY,IM),JY=1,MY)
C      CALCULATION OF A(1),A(2),A(3)
C      IN DY(MY)=A(1)*(1.0-Y(MY))**A(3)*(Y(MY)+A(2)).
       WRITE(6,54)
       MM=1
   315 DO 140 JX=1,MX
       DO 145 I=1,N
   145 X(JX,I)=XB(JX,I,IM)
       XA(JX)=XBA(JX,IM)
       XE(JX)=XBE(JX,IM)
   140 HX(JX)=(XBE(JX,IM)-XA(JX))/200.0
       DO 150 JY=1,MY
       DO 155 I=1,N
       Y(JY,I)=YB(JY,I,IM)
   155 W(JY,I)=1.0
       DO 160 JX=1,MX
   160 DYA(JY,JX)=(Y(JY,1)-YA(JY))/(X(JX,1)-XA(JX))
       YA(JY)=YBA(JY,IM)
   150 YE(JY)=YBE(JY,IM)
       GO TO (195,320,410,515,610), MM
   195 L=3
C      CALCULATION OF INITIAL VALUES OF A(1),A(2),A(3).
   520 LL=2
       II=0
       NN=0
   200 II=II+1
       IF(N-II) 205,210,210
   210 IF(Y(1,II)-0.2) 200,215,215
   215 IF(0.8-Y(1,II)) 200,220,220
   220 NN=NN+1
       YYY(NN)=Y(1,II)
       XX(1,NN)=1.0
       XX(2,NN)=X(1,II)
       WW(NN)=1.0
       GO TO 200
   205 CALL SENSAS(YYY,XX,AAA,YYC,WW,SSD,NN,NNT,LL,LLT,ILL)
       IF(ILL) 225,230,225
   225 WRITE(6,55) ILL
       GO TO 105
   230 GO TO (525,525,525,530), MM
```

```
  525 IF(AAA(1)) 235,240,240
  235 A(2,1)=1.0/EXP(2.0-4.0*AAA(1))
      A(1,1)=4.0*AAA(2)/(1.0+A(2,1))**2
      A(3,1)=1.0
      GO TO 245
  240 AS=0.0
      DO 250 II=1,NN
  250 AS=AS+YYY(II)/XX(1,II)
      A(1,1)=AS/NN
      A(2,1)=1.0
      A(3,1)=1.0
  245 WRITE(6,56) (A(IL,1),IL=1,L)
      HM(1)=1.0
      HS(1)=1.0
      CALL HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS
     1,DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY
     2,AA,Q,DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQA,ILL)
      IF(ILL) 225,165,225
  165 WRITE(6,57) L,K
      WRITE(6,58)
      WRITE(6,59) HM(K1),HS(K1),(A(IL,K1),IL=1,L),SD(K1)
      WRITE(6,59) HM(K),HS(K),(A(IL,K),IL=1,L),SD(K)
      WRITE(6,60)
      DO 170 JY=1,MY
      WRITE(6,59) (YC(JY,I,K1),I=1,N)
      WRITE(6,59) (YC(JY,I,K),I=1,N)
      WRITE(6,59) (DYC(JY,I,K1),I=1,N)
      WRITE(6,59) (DYC(JY,I,K),I=1,N)
  170 CONTINUE
      DO 175 IL=1,L
  175 AB(IL,IM)=A(IL,K)
      GO TO (115,300,400,500,600), MM
  120 SAM=0.0
      DO 180 IM=1,M
  180 SAM=SAM+AB(2,IM)
      AMAV=SAM/M
      IF(AMAV-0.75) 185,190,190
  190 IM=0
      WRITE(6,50) M
      MM=MM+1
  300 IM=IM+1
      IF(M-IM) 305,310,310
  310 N=NM(IM)
C     CALCULATION OF A(1),A(2) IN DY(MY)=A(1)*(1.0-Y(MY))**A(2).
      WRITE(6,61)
      GO TO 315
  320 L=2
C     CALCULATION OF INITIAL VALUES OF A(1),A(2)
      AS=0.0
      DO 325 I=1,N
  325 AS=AS-ALOG(1.0-Y(1,I))/X(1,I)
      A(1,1)=AS/N
      A(2,1)=1.0
      WRITE(6,56) (A(IL,1),IL=1,L)
      HM(1)=1.0
      HS(1)=1.0
      CALL HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS
     1,DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY
     2,AA,Q,DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQB,ILL)
      IF(ILL) 225,165,225
  305 SAN=0.0
      DO 330 IM=1,M
  330 SAN=SAN+AB(2,IM)
      ANAV=SAN/M
      IF(ANAV-0.75) 335,340,340
  335 AN=0.5
      GO TO 345
  340 IF(ANAV-1.25) 350,355,355
  350 AN=1.0
      GO TO 345
  355 IF(ANAV-1.75) 360,365,365
  360 AN=1.5
      GO TO 345
  365 AN=2.0
  345 IM=0
      WRITE(6,50) M
      MM=MM+1
  400 IM=IM+1
      IF(M-IM) 100,405,405
  405 N=NM(IM)
C     CALCULATION OF A(1) IN DY(MY)=A(1)*(1.0-Y(MY))**AN.
      WRITE(6,62) AN
      GO TO 315
```

```
  410 L=1
C     CALCULATION OF INITIAL VALUES OF A(1).
      IF(AN-1) 420,425,420
  425 AS=0.0
      DO 430 I=1,N
  430 AS=AS-ALOG(1.0-Y(1,I))/X(1,I)
      A(1,1)=AS/N
      GO TO 435
  420 IF(AN-0.5) 440,445,440
  445 AS=0.0
      DO 450 I=1,N
  450 AS=AS-2.0*(SQRT(1.0-Y(1,I))-1.0)/X(1,I)
      A(1,1)=AS/N
      GO TO 435
  440 AS=0.0
      DO 455 I=1,N
  455 AS=AS+(1.0/(1.0-Y(1,I))**(AN-1.0)-1.0)/((AN-1.0)*X(1,I))
      A(1,1)=AS/N
  435 WRITE(6,56) (A(IL,1),IL=1,L)
      HM(1)=1.0
      HS(1)=1.0
      CALL HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS
     1,DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY
     2,AA,Q,DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQC,ILL)
      IF(ILL) 225,165,225
  185 IM=0
      WRITE(6,50) M
      MM=MM+3
  500 IM=IM+1
      IF(M-IM) 505,510,510
  510 N=NM(IM)
C     CALCULATION OF A(1),A(2) IN DY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+A(2)).
      WRITE(6,63)
      GO TO 315
  515 L=2
C     CALCULATION OF INITIAL VALUES OF A(1),A(2).
      GO TO 520
  530 IF(AAA(1)-0.5) 535,540,540
  535 A(2,1)=1.0/EXP(2.0-4.0*AAA(1))
      A(1,1)=4.0*AAA(2)/(1.0+A(2,1))**2
      GO TO 545
  540 AS=0.0
      DO 550 II=1,NN
  550 AS=AS+YYY(II)/XX(1,II)
      A(1,1)=AS/NN
      A(2,1)=1.0
  545 WRITE(6,56) (A(IL,1),IL=1,L)
      HM(1)=1.0
      HS(1)=1.0
      CALL HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS
     1,DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY
     2,AA,Q,DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQD,ILL)
      IF(ILL) 225,165,225
  505 SAM=0.0
      DO 555 IM=1,M
  555 SAM=SAM+AB(2,IM)
      AMAV=SAM/M
      IF(AMAV-0.25) 560,565,565
  565 AM=0.5
      GO TO 570
  560 IF(AMAV-0.05) 575,580,580
  580 AM=0.1
      GO TO 570
  575 IF(AMAV-0.005) 585,590,590
  590 AM=0.01
      GO TO 570
  585 AM=0.001
  570 IM=0
      WRITE(6,50) M
      MM=MM+1
  600 IM=IM+1
    . IF(M-IM) 100,605,605
  605 N=NM(IM)
C     CALCULATION OF A(1) IN DM(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+AM).
      WRITE(6,64) AM
      GO TO 315
  610 L=1
C     CALCULATION OF INITIAL VALUES OF A(1).
      AS=0.0
      DO 615 I=1,N
  615 AS=AS+4.0*(Y(1,I)-(0.5*(1.0-AM)+0.25*(1.0+AM)*ALOG(AM)))
     1/((1.0+AM)**2*X(1,I))
      A(1,1)=AS/N
```

```
      WRITE(6,56) (A(IL,1),IL=1,L)
      HM(1)=1.0
      HS(1)=1.0
      CALL HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS
     1,DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY
     2,AA,Q,DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQE,ILL)
      IF(ILL) 225,165,225
  105 STOP
   10 FORMAT(10I4)
   15 FORMAT(10F8.0)
   50 FORMAT(1H1,2HM=,5X,I4)
   51 FORMAT(1H0,3HMY=,I4,5X,3HMX=,I4,5X,2HN=,I4)
   52 FORMAT(1H0,31HYBA(MY,M),YB(MY,N,M),YBE(MY,M)=/(1H ,5E13.5))
   53 FORMAT(1H0,31HXBA(MX,M),XB(MX,N,M),XBE(MX,M)=/(1H ,5E13.5))
   54 FORMAT(1H0,42HDY(MY)=A(1)*(1.0-Y(MY))**A(3)*(Y(MY)+A(2)))
   55 FORMAT(1H1,4HILL=,I8)
   56 FORMAT(1H0,7HA(L,0)=/(1H ,5E13.5))
   57 FORMAT(1H0,2HL=,I4,5X,6HK=    0,5X,2HK=,I4)
   58 FORMAT(1H ,25HHM(K),HS(K),A(L,K),SD(K)=)
   59 FORMAT(1H ,5E13.5)
   60 FORMAT(1H0,11HYC(MY,N,K)=/1H ,12HDYC(MY,N,K)=)
   61 FORMAT(1H0,29HDY(MY)=A(1)*(1.0-Y(MY))**A(2))
   62 FORMAT(1H0,27HDY(MY)=A(1)*(1.0-Y(MY))**AN/1H ,3HAN=,F8.4)
   63 FORMAT(1H0,36HDY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+A(2)))
   64 FORMAT(1H0,34HDY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+AM)/1H ,3HAM=,F8.4)
      END
      SUBROUTINE SENSAS(Y,X,A,YC,W,SD,N,NT,L,LT,ILL)
C     METHOD OF LINEAR LEAST SQUARE
      DIMENSION Y(NT),X(LT,NT),A(LT),YC(NT),W(NT),T(20,21)
      IF(N.GE.L.OR.L.GE.2.OR.L.LE.20) GO TO 20
      ILL=30000
      GO TO 999
   20 LP=L+1
      LM=L-1
      DO 100 I=1,L
      DO 200 J=1,LP
  200 T(I,J)=0.0
      DO 100 K=1,N
      DO 300 J=1,L
  300 T(I,J)=T(I,J)+X(I,K)*X(J,K)*W(K)**2
  100 T(I,L+1)=T(I,L+1)+X(I,K)*Y(K)*W(K)**2
      DO 400 I=1,LM
      IF(T(I,I)) 30,40,30
   40 ILL=I
      GO TO 999
   30 IP=I+1
      DO 400 J=IP,L
      DO 400 K=J,LP
  400 T(J,K)=T(J,K)-(T(I,J)*T(I,K))/T(I,I)
      DO 500 I=1,L
      J=L+1-I
      K=J
      AT=0.0
   60 IF(L-K) 500,500,50
   50 AT=AT+A(K+1)*T(J,K+1)
      K=K+1
      GO TO 60
  500 A(J)=(T(J,L+1)-AT)/T(J,J)
      DO 600 I=1,N
      YC(I)=0.0
      DO 600 J=1,L
  600 YC(I)=YC(I)+A(J)*X(J,I)*W(I)
      D=0.0
      DO 700 I=1,N
  700 D=D+(Y(I)*W(I)-YC(I))**2
      SD=SQRT(D/N)
      ILL=0
  999 RETURN
      END
      SUBROUTINE HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,DYC,YC,YCS,
     1DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,DYY,YY,AA,Q,
     2DYS,YS,DYA,YA,YE,XA,XE,HX,DIFEQ,ILL)
C     METHOD OF NONLINEAR LEAST SQUARE
      DIMENSION Y(MYT,NT),X(MXT,NT),A(LT,K1),HM(K1),HS(K1),W(MYT,NT),
     1DYC(MYT,NT,K1),YC(MYT,NT,K1),YCS(MYT,NT,L1),DD(LT,L1),SD(K1),
     2HA(20),AS(20),DDS(20),DYY(MYT),YY(MYT),AA(LT),Q(MYT),DYS(MYT),
     3YS(MYT),DYA(MYT,MXT),YA(MYT),YE(MYT),XA(MXT),XE(MXT),HX(MXT)
      IF(KT.GE.1.OR.MY.GE.1.OR.MX.GE.1.OR.N.GE.L.OR.L.GE.1.OR
     1.L.LE.20.OR.L.LT.L1.OR.KT.LT.K1.OR.EPS.GE.0.0) GO TO 200
      ILL=30000
      GO TO 999
  200 K=1
      HS(K1)=HS(1)
```

```
      HM(K1)=HM(1)
      SDS=0.0
      DO 10 IA=1,L
   10 A(IA,K1)=A(IA,1)
  888 CALL SUB(DYC,YC,X,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,DYY,YY,
     1AA,Q,DYS,YS,DYA,YA,XA,HX,DIFEQ,ILL)
      IF(ILL) 300,400,300
  300 IF(SDS) 999,999,700
  400 D=0.0
      DO 20 I=1,N
      DO 20 J=1,MY
   20 D=D+((Y(J,I)-YC(J,I,K))*W(J,I))**2
      SD(K)=SQRT(D/(N*MY))
      IF(SDS) 999,500,600
  500 SDS=SD(K)
      SD(K1)=SD(K)
      DO 30 I=1,N
      DO 30 J=1,MY
      YCS(J,I,L1)=YC(J,I,K)
      DYC(J,I,K1)=DYC(J,I,K)
   30 YC(J,I,K1)=YC(J,I,K)
      DO 21 IA=1,L
   21 A(IA,K1)=A(IA,1)
      GO TO 777
  600 IF(SDS-SD(K)) 700,800,800
  700 HS(K)=HS(K)*HSS
      IF(HST-HS(K)) 666,999,999
  800 IF(KT-K) 999,999,900
  900 K=K+1
      HM(K)=HM(K-1)*HMM
      HS(K)=HS(1)
      SDS=SD(K-1)
      DO 40 IA=1,L
   40 A(IA,K)=A(IA,K-1)
      DO 50 I=1,N
      DO 50 J=1,MY
   50 YCS(J,I,L1)=YC(J,I,K-1)
  777 DO 60 IA=1,L
      AST=A(IA,K)
      HA(IA)=A(IA,K)*HAA
      A(IA,K)=A(IA,K)+HA(IA)
      CALL SUB(DYC,YC,X,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,DYY,YY,
     1AA,Q,DYS,YS,DYA,YA,XA,HX,DIFEQ,ILL)
      IF(ILL) 999,1000,999
 1000 DO 70 I=1,N
      DO 70 J=1,MY
      YCS(J,I,IA)=YC(J,I,K)
   70 YCS(J,I,IA)=(YCS(J,I,IA)-YCS(J,I,L1))/HA(IA)
   60 A(IA,K)=AST
      LP=L+1
      DO 80 IA1=1,L
      DO 80 IA2=1,L
      DD(IA1,IA2)=0.0
      DO 80 I=1,N
      DO 80 J=1,MY
   80 DD(IA1,IA2)=DD(IA1,IA2)+YCS(J,I,IA1)*YCS(J,I,IA2)
     1*W(J,I)**2
      DO 90 IA=1,L
      DD(IA,IA)=DD(IA,IA)*(1.0+HM(K))
      DD(IA,LP)=0.0
      DO 90 I=1,N
      DO 90 J=1,MY
   90 DD(IA,LP)=DD(IA,LP)+YCS(J,I,IA)*(Y(J,I)-YCS(J,I,L1))
     1*W(J,I)**2
      IF(L-1) 999,333,222
  222 CALL GAUYOS(DD,L,LP,LT,L1,EPS,ILL)
      IF(ILL) 999,1100,999
  333 DD(1,2)=DD(1,2)/DD(1,1)
      ILL=0
 1100 DO 110 IA=1,L
      AS(IA)=A(IA,K)
  110 DDS(IA)=DD(IA,LP)
  666 DO 120 IA=1,L
      DD(IA,LP)=DDS(IA)*HS(K)
  120 A(IA,K)=AS(IA)+DD(IA,LP)
      GO TO 888
  999 RETURN
      END

      SUBROUTINE SUB(DYC,YC,X,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,DYY,
     1YY,AA,Q,DYS,YS,DYA,YA,XA,HX,DIFEQ,ILL)
C     CALCULATION OF EQUATIONS FOR SIMULATION
      DIMENSION DYC(MYT,NT,K1),YC(MYT,NT,K1),X(MXT,NT),A(LT,K1),
```

```
     1DYY(MYT),YY(MYT),AA(LT),Q(MYT),DYS(MYT),YS(MYT),DYA(MYT,MXT)
     2,YA(MYT),XA(MXT),HX(MXT)
      XX=XA(1)
      HXX=HX(1)
      DO 100 I=1,L
  100 AA(I)=A(I,K)
      DO 200 I=1,MY
      DYY(I)=DYA(I,1)
      YY(I)=YA(I)
  200 Q(I)=0.0
      DO 300 I=1,N
      IF(XX-X(1,I)) 400,500,500
  400 DO 600 J=1,MY
      DYS(J)=DYY(J)
  600 YS(J)=YY(J)
      CALL URKGS(DYY,YY,XX,AA,Q,HXX,MY,MYT,L,LT,DIFEQ,ILL)
      IF(ILL) 999,700,999
  700 IF(XX-X(1,I)) 400,500,500
  500 DO 300 J=1,MY
      DYC(J,I,K)=DYY(J)-(DYY(J)-DYS(J))*(XX-X(1,I))/HXX
  300 YC(J,I,K)=YY(J)-(YY(J)-YS(J))*(XX-X(1,I))/HXX
  999 RETURN
      END
      SUBROUTINE GAUYOS(A,N,N1,NT,NT1,EPS,ILL)
C     GAUSS-JORDAN METHOD
      DIMENSION A(NT,NT1)
      DO 10 K=1,N
      BIG=ABS(A(K,K))
      IP=K
      K1=K+1
      IF(K1.GT.N) GO TO 14
      DO 11 I=K1,N
      IF(ABS(A(I,K)).LE.BIG) GO TO 11
      BIG=ABS(A(I,K))
      IP=I
   11 CONTINUE
   14 IF(BIG.GE.EPS) GO TO 12
      ILL=1000
      GO TO 999
   12 IF(IP.EQ.K) GO TO 15
      DO 13 J=1,N1
      TEMP=A(K,J)
      A(K,J)=A(IP,J)
   13 A(IP,J)=TEMP
   15 W=A(K,K)
      DO 20 J=K1,N1
   20 A(K,J)=A(K,J)/W
      DO 30 I=1,N
      IF(I.EQ.K) GO TO 30
      W=A(I,K)
      DO 40 J=K1,N1
   40 A(I,J)=A(I,J)-W*A(K,J)
   30 CONTINUE
   10 CONTINUE
      ILL=0
  999 RETURN
      END
      SUBROUTINE URKGS(DY,Y,X,A,Q,HX,MY,MYT,L,LT,DIFEQ,ILL)
C     RUNGE-KUTTA-GILL METHOD
      DIMENSION DY(MYT),Y(MYT),A(LT),Q(MYT),T(20),R(20),P(6),B(4),C(4)
      DATA P(1),C(1),C(4)/3*0.0/,B(1)/1.0/
      DATA B(2)/0.2928932/,B(3)/1.707107/
      PX=X
      P(2)=0.5*HX
      P(3)=0.5*HX
      P(6)=0.5*HX
      P(4)=HX
      P(5)=HX
      B(4)=1.0/3.0
      C(2)=0.7071068*HX
      C(3)=-C(2)
      DO 100 J=1,4
      X=PX+P(J)
      CALL DIFEQ(DY,Y,X,A,MY,MYT,L,LT,ILL)
      IF(ILL) 999,200,999
  200 DO 100 I=1,MY
      T(I)=P(J+2)*DY(I)-Q(I)
      R(I)=B(J)*T(I)
      Y(I)=Y(I)+R(I)
  100 Q(I)=3.0*R(I)-T(I)+C(J)*DY(I)
  999 RETURN
      END
```

```
      SUBROUTINE DIFEQA(DY,Y,X,A,MY,MYT,L,LT,ILL)
C     CALCULATION OF ORDINARY DIFFERENTIAL EQUATIONS.
C     DY(MY)=,  DIFFERENTIAL VALUES OF Y(MY) BY X.
C     Y(MY),X=,  DEPENDENT AND INDEPENDENT VARIABLES.
C     A(L)=,  RATE PARAMETERS.
C     CALCULATION OF DY(MY)=A(1)*(1.0-Y(MY))**A(3)*(Y(MY)+A(2))
      DIMENSION DY(MYT),Y(MYT),A(LT)
      AA=1.0-Y(1)
      IF(AA) 100,200,200
  100 AA=1.0E-10
  200 DY(1)=A(1)*AA**A(3)*(Y(1)+A(2))
      ILL=0
      RETURN
      END
      SUBROUTINE DIFEQB(DY,Y,X,A,MY,MYT,L,LT,ILL)
C     CALCULATION OF DY(MY)=A(1)*(1.0-Y(MY))**A(2)
      DIMENSION DY(MYT),Y(MYT),A(LT)
      AA=1.0-Y(1)
      IF(AA) 100,200,200
  100 AA=1.0E-10
  200 DY(1)=A(1)*AA**A(2)
      ILL=0
      RETURN
      END
      SUBROUTINE DIFEQC(DY,Y,X,A,MY,MYT,L,LT,ILL)
C     CALCULATION OF DY(MY)=A(1)*(1.0-Y(MY))**AN
      COMMON AN,AM
      DIMENSION DY(MYT),Y(MYT),A(LT)
      AA=1.0-Y(1)
      IF(AA) 100,200,200
  100 AA=1.0E-10
  200 DY(1)=A(1)*AA**AN
      ILL=0
      RETURN
      END
      SUBROUTINE DIFEQD(DY,Y,X,A,MY,MYT,L,LT,ILL)
C     CALCULATION OF DY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+A(2))
      DIMENSION DY(MYT),Y(MYT),A(LT)
      AA=1.0-Y(1)
      IF(AA) 100,200,200
  100 AA=1.0E-10
  200 DY(1)=A(1)*AA*(Y(1)+A(2))
      ILL=0
      RETURN
      END
      SUBROUTINE DIFEQE(DY,Y,X,A,MY,MYT,L,LT,ILL)
C     CALCULATION OF DY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+AM)
      COMMON AN,AM
      DIMENSION DY(MYT),Y(MYT),A(LT)
      AA=1.0-Y(1)
      IF(AA) 100,200,200
  100 AA=1.0E-10
  200 DY(1)=A(1)*AA*(Y(1)+AM)
      ILL=0
      RETURN
      END
```

```
M=        1

MY=   1      MX=   1      N=   11

YBA(MY,M),YB(MY,N,M),YBE(MY,M)=
  0.15000E 01  0.14750E 01  0.14560E 01  0.14000E 01  0.13500E 01
  0.13000E 01  0.12500E 01  0.12000E 01  0.11500E 01  0.11000E 01
  0.10500E 01  0.10250E 01  0.10000E 01

XBA(MX,M),XB(MX,N,M),XBE(MX,M)=
  0.00000E 00  0.60800E 01  0.82600E 01  0.10800E 02  0.12500E 02
  0.13900E 02  0.15400E 02  0.16600E 02  0.18000E 02  0.19800E 02
  0.22500E 02  0.25000E 02  0.50000E 02

YBA(MY,M),YB(MY,N,M),YBE(MY,M)=
  0.00000E 00  0.49999E-01  0.87999E-01  0.20000E 00  0.30000E 00
  0.40000E 00  0.50000E 00  0.60000E 00  0.70000E 00  0.80000E 00
  0.90000E 00  0.95000E 00  0.10000E 01

DY(MY)=A(1)*(1.0-Y(MY))**A(3)*(Y(MY)+A(2))

A(L,0)=
  0.26651E 00  0.15036E-01  0.10000E 01
```

```
L=   3      K=   0      K=  20
HM(K),HS(K),A(L,K),SD(K)=
  0.10000E 01  0.10000E 01  0.26651E 00  0.15036E-01  0.10000E 01
  0.26170E-01
  0.19073E-05  0.78125E-02  0.30629E 00  0.91506E-02  0.10134E 01
  0.45424E-02

YC(MY,N,K)=
DYC(MY,N,K)=
  0.58337E-01  0.10999E 00  0.20657E 00  0.29623E 00  0.38333E 00
  0.48455E 00  0.56637E 00  0.65688E 00  0.75752E 00  0.86671E 00
  0.92755E 00
  0.47911E-01  0.96905E-01  0.19725E 00  0.29620E 00  0.39472E 00
  0.50933E 00  0.60031E 00  0.69755E 00  0.79948E 00  0.90003E 00
  0.94997E 00
  0.18413E-01  0.29656E-01  0.46855E-01  0.58383E-01  0.65455E-01
  0.68612E-01  0.67174E-01  0.61444E-01  0.49918E-01  0.31322E-01
  0.18201E-01
  0.16628E-01  0.29295E-01  0.50591E-01  0.65514E-01  0.74345E-01
  0.77153E-01  0.73673E-01  0.64424E-01  0.48597E-01  0.26990E-01
  0.14117E-01
M=         1

DY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+A(2))

A(L,0)=
  0.26651E 00  0.15036E-01

L=   2      K=   0      K=  16
HM(K),HS(K),A(L,K),SD(K)=
  0.10000E 01  0.10000E 01  0.26651E 00  0.15036E-01  0.26170E-01
  0.30518E-04  0.61035E-04  0.30266E 00  0.95398E-02  0.46147E-02

YC(MY,N,K)=
DYC(MY,N,K)=
  0.58337E-01  0.10999E 00  0.20657E 00  0.29623E 00  0.38333E 00
  0.48455E 00  0.56637E 00  0.65688E 00  0.75752E 00  0.86671E 00
  0.92755E 00
  0.48659E-01  0.97841E-01  0.19794E 00  0.29636E 00  0.39437E 00
  0.50862E 00  0.59960E 00  0.69718E 00  0.79985E 00  0.90131E 00
  0.95149E 00
  0.18413E-01  0.29656E-01  0.46855E-01  0.58383E-01  0.65455E-01
  0.68612E-01  0.67174E-01  0.61444E-01  0.49919E-01  0.31322E-01
  0.18201E-01
  0.16756E-01  0.29320E-01  0.50359E-01  0.65146E-01  0.74011E-01
  0.77034E-01  0.73793E-01  0.64771E-01  0.49023E-01  0.27205E-01
  0.14109E-01
M=         1

DY(MY)=A(1)*(1.0-Y(MY))*(Y(MY)+AM)
AM=  0.0100

A(L,0)=
  0.31401E 00

L=   1      K=   0      K=  20
HM(K),HS(K),A(L,K),SD(K)=
  0.10000E 01  0.10000E 01  0.31401E 00  0.39401E-01
  0.19073E-05  0.25000E 00  0.29964E 00  0.48547E-02

YC(MY,N,K)=
DYC(MY,N,K)=
  0.55029E-01  0.11195E 00  0.22745E 00  0.33850E 00  0.44547E 00
  0.56493E 00  0.65567E 00  0.74841E 00  0.84052E 00  0.92552E 00
  0.96488E 00
  0.49860E-01  0.99667E-01  0.20017E 00  0.29834E 00  0.39571E 00
  0.50898E 00  0.59914E 00  0.69593E 00  0.79807E 00  0.89962E 00
  0.95027E 00
  0.19295E-01  0.34005E-01  0.57592E-01  0.72390E-01  0.79280E-01
  0.78515E-01  0.71949E-01  0.59914E-01  0.42585E-01  0.21878E-01
  0.10752E-01
  0.17041E-01  0.29585E-01  0.50363E-01  0.64828E-01  0.73438E-01
  0.76331E-01  0.73142E-01  0.64318E-01  0.48886E-01  0.27358E-01
  0.14308E-01
```

各種食品の各種化学的，物理的処理における経験的速度式の設定に関する研究

久保田 清

各種食品の化学反応，乾燥，蒸煮，抽出装置など各種化学的，物理的処理装置を設計し，制御化などを行なっていくためには，最適な各種処理速度式を設定していくことが必要である。一般に，食品の各種処理機構は明確でなく，理論的速度式を得ることが困難であるため，経験的速度式の検討が必要となり，その設定方法が問題となる。

本研究は，各種食品の各種化学的，物理的処理における経験的速度式の設定方法について研究を行ない，電子計算機プログラムの作成を行なったものである。本研究で得られた経験的速度式の設定のためのプログラムは，実験データが単調な増減をする場合と，S字型の形状を示して増減をする場合との両方に対して有用なものである。また，本研究で得られたプログラムは，各種食品以外のものに対しても，経験的速度式を設定するのに有用である。